
Change 1 –  f_IMS_InviteRequest_MessageHeaderRX
	Function name
	 f_IMS_InviteRequest_MessageHeaderRX

	Reason for change
	34.229-1, A.2.1, does not list the lr parameter for conditions A6 and A7 regarding the iniation of emergency calls. However, TTCN requires the UE to bring the parameter.
Note: After bringing TTCN in line with 34.229-1, we need to bring the issue to RAN5 attendance as absence of the lr parameter might be in contradiction to core specifications.

	Summary of change
	Used the template cr_SipUri_HostPort instead of cr_SipUri_HostPort_lr such that UE is not required anymore to bring the lr parameter  

	TTCN module
	Common/IMS/IMS_Procedures_CallControl 

	MCC160 Comment
	


Before change

  function f_IMS_InviteRequest_MessageHeaderRX(template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                               template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := pc_IMS_SessionIdSupported,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  { /* p_A3:          MTSI

       p_A8:          UE uses Geolocation header to provide its geographical location for emergency session setup, has obtained its location and is setting up an emergency session

       p_A9:          UE supports Session-ID   @sic R5s150086: pc_IMS_SessionIdSupported per default sic@

       p_A11:         INVITE for creating a video call and UE supports video media feature tag

       NOTE1: If SigComp shall be applied the param "comp=sigcomp" is optional for Via, Route and Contact

       NOTE2: condition A5 (re-INVITE within a dialog) is not considered yet and will have impact e.g. on checking of the From/To headers */

    /* @sic R5-133629: 100rel option tag shall always be included in INVITE; further option tags may be added => p_Supported replaced by p_AdditionalOptionTagsForSupported; Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5s140017 MCC160 implementation: checks for A_2_1_A5 sic@ */

    /* @sic R5-140799: additional parameter p_MessageHeader to deal with Reason header sic@ */

    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2

    var template (present) MessageHeader v_MessageHeader_Invite := p_MessageHeader;

    var template (present) RouteBody_List v_RouteBodyList;

    var IMS_ProtectedPorts_Type v_Protected;

    var template integer v_Port_us := *;

    var template integer v_Port_ps := *;

    var template DisplayName v_DisplayName := *;

    var template (present) SipUrl v_SipUrlFrom := ?;    /* per default any SIP URI is accepted in the From header in the first place;

                                                           further checks may be applied in f_IMS_A_2_1_Invite_CommonChecks (depending on the conditions acc. to A.2.1 */

    var template (present) charstring v_Pcscf;

    var template (present) charstring v_MediaType := "application/sdp";

    var template (present) CallId v_CallId;

    var template (present) integer v_CSeqValue := ?;

    var template (present) From v_From;

    var template (present) To v_To;

    var template SessionId v_SessionId;

    var template AcceptBody_List v_AcceptBodyList := {};

    var OptionTag_List v_OptionTag_List := tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported;

    if (pc_IMS_CS_PS_SRVCCMidCall or pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@

      v_OptionTag_List := v_OptionTag_List & tsc_OptionTagList_norefersub;

    }

    if (p_Context == A_2_1_A6) {  // INVITE for creating an emergency session in case of no registration

      // Via, Route, Contact: ports to be used

      v_Port_us := ?;

      v_Port_ps := tsc_IMS_PortNumber_5060;

      // From: Display

      v_DisplayName := cr_DisplayName("Anonymous");     /* @sic R5s140005, R5-140901, R5w140111 sic@ */

    }

    else {

      // Via, Route, Contact: ports to be used

      if (not v_IsGiba or (p_Context == A_2_1_A7)) {

        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

        v_Port_us := v_Protected.Port_us;

        v_Port_ps := v_Protected.Port_ps;

      }

    }

    if (p_Context == A_2_1_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */

      v_RouteBodyList := f_RouteSet_RouteBodyList_RX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      v_From := f_IMS_RequestInDialog_FromHeaderRX();

      v_To := f_IMS_RequestInDialog_ToHeaderRX();

      v_CSeqValue := f_IMS_RequestInDialog_CSeqValueRX();     /* @sic R5s140235, R5s140385 change 3 MCC160 implementation sic@ */

      v_SessionId := f_IMS_PTC_ImsInfo_DialogGetSessionId();

    }

    else {                                       /* normal case: no re-invite */

      // route body acc. to conditions in A.2.1

      v_Pcscf := ?;                                                                      /* @sic R5s140934 change 6: P-CSCF address is checked later on sic@ */

      v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(v_Pcscf, v_Port_ps));  /* @sic R5s140873: v_Pcscf;  R5s140934 change 6 sic@ */
      if (not match(p_Context, (A_2_1_A6, A_2_1_A7))) {  // => no emergency call

        v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Scscf));

      }

      if (p_A8) {

        v_MessageHeader_Invite.geolocation := cr_GeolocationAny;

        v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

        v_MediaType := "multipart/mixed";

      }

      v_CallId := cr_CallId(?);

      v_From := cr_From(v_DisplayName, v_SipUrlFrom);

      v_To := cr_ToDef(p_ExpectedCalleeUri);

      v_SessionId := f_IMS_SessionId_RX_IfSupported(p_A9);

    }

    v_MessageHeader_Invite.via             := f_IMS_InviteRequest_ViaRX();                            /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address

                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */

    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := v_From;                                                 /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */

    v_MessageHeader_Invite.toField         := v_To;                                                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */

    v_MessageHeader_Invite.callId          := v_CallId;                                               /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */

    v_MessageHeader_Invite.sessionId       := v_SessionId;                                            /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Invite.cSeq            := cr_CseqDef(v_CSeqValue, "INVITE");                      /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */

    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);

    v_MessageHeader_Invite.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us));           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check of feature-param for A3, A6, A10, A12 */

    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);

    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;

    // additional headers for specific conditions:

    if (v_IsGiba or (p_Context == A_2_1_A6)) {

      v_MessageHeader_Invite.securityVerify  := omit;

      if (p_Context == A_2_1_A6) {

        v_MessageHeader_Invite.require       := omit;

        v_MessageHeader_Invite.proxyRequire  := omit;

      }

    } else {   // -> A1 or A7

      v_MessageHeader_Invite.securityVerify  := f_IMS_PTC_BuildSecurityVerifyHeader();

      if (p_Context != A_2_1_A5) {  /* @sic R5s140017 change 9.3 sic@ */

        v_MessageHeader_Invite.require       := cr_RequireSecAgree;

        v_MessageHeader_Invite.proxyRequire  := cr_ProxyRequireDef;

      }

    }

    if (not v_IsGiba) {

      v_MessageHeader_Invite.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString());

    }

    if (p_Context == A_2_1_A4) {  /* @sic R5s140017 change 9.3 sic@ */

      v_AcceptBodyList := {cr_AcceptBody("application/sdp"), cr_AcceptBody("application/3gpp-ims+xml")};

    }

    if (pc_IMS_CS_PS_SRVCCMidCall) { // @sic R5-145746 sic@

      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.mid-call+xml"); // add another element in the array

    }

    if (pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@

      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.state-and-event-info+xml"); // add another element in the array

      v_MessageHeader_Invite.recvInfo        := cr_RecvInfo ({cr_InfoPackage_Type("+g.3gpp.state-and-event")});  // @sic R5-145746 sic@

    }

    if (lengthof(v_AcceptBodyList) > 0) {

      v_MessageHeader_Invite.accept := cr_Accept(superset(all from v_AcceptBodyList));

    }

    if (p_A8) {

      v_MessageHeader_Invite.geolocation     := cr_GeolocationAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks @sic R5-133151 sic@

    }

    if (p_Context == A_2_1_A7) {

      v_MessageHeader_Invite.pPreferredID := cr_PPreferredID(?);    // Checked in f_IMS_A_2_1_Invite_CommonChecks

    }

    if (p_A3 and (p_Context == A_2_1_A4)) {                          /* @sic R5s140263 change 5, modified by R5-142949 (-> A4 only) sic@ */

      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_FeatureParamValue_3gpp_icsi_ref);   /* @sic R5s140565 change 2 sic@ */

    }

    if ((p_A3 and (p_Context == A_2_1_A4)) or p_A11) {               /* @sic R5-142949 sic@ */

      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)

    }

    return v_MessageHeader_Invite;

  }      
After change

  function f_IMS_InviteRequest_MessageHeaderRX(template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                               template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := pc_IMS_SessionIdSupported,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  { /* p_A3:          MTSI

       p_A8:          UE uses Geolocation header to provide its geographical location for emergency session setup, has obtained its location and is setting up an emergency session

       p_A9:          UE supports Session-ID   @sic R5s150086: pc_IMS_SessionIdSupported per default sic@

       p_A11:         INVITE for creating a video call and UE supports video media feature tag

       NOTE1: If SigComp shall be applied the param "comp=sigcomp" is optional for Via, Route and Contact

       NOTE2: condition A5 (re-INVITE within a dialog) is not considered yet and will have impact e.g. on checking of the From/To headers */

    /* @sic R5-133629: 100rel option tag shall always be included in INVITE; further option tags may be added => p_Supported replaced by p_AdditionalOptionTagsForSupported; Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5s140017 MCC160 implementation: checks for A_2_1_A5 sic@ */

    /* @sic R5-140799: additional parameter p_MessageHeader to deal with Reason header sic@ */

    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2

    var template (present) MessageHeader v_MessageHeader_Invite := p_MessageHeader;

    var template (present) RouteBody_List v_RouteBodyList;

    var IMS_ProtectedPorts_Type v_Protected;

    var template integer v_Port_us := *;

    var template integer v_Port_ps := *;

    var template DisplayName v_DisplayName := *;

    var template (present) SipUrl v_SipUrlFrom := ?;    /* per default any SIP URI is accepted in the From header in the first place;

                                                           further checks may be applied in f_IMS_A_2_1_Invite_CommonChecks (depending on the conditions acc. to A.2.1 */

    var template (present) charstring v_Pcscf;

    var template (present) charstring v_MediaType := "application/sdp";

    var template (present) CallId v_CallId;

    var template (present) integer v_CSeqValue := ?;

    var template (present) From v_From;

    var template (present) To v_To;

    var template SessionId v_SessionId;

    var template AcceptBody_List v_AcceptBodyList := {};

    var OptionTag_List v_OptionTag_List := tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported;

    if (pc_IMS_CS_PS_SRVCCMidCall or pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@

      v_OptionTag_List := v_OptionTag_List & tsc_OptionTagList_norefersub;

    }

    if (p_Context == A_2_1_A6) {  // INVITE for creating an emergency session in case of no registration

      // Via, Route, Contact: ports to be used

      v_Port_us := ?;

      v_Port_ps := tsc_IMS_PortNumber_5060;

      // From: Display

      v_DisplayName := cr_DisplayName("Anonymous");     /* @sic R5s140005, R5-140901, R5w140111 sic@ */

    }

    else {

      // Via, Route, Contact: ports to be used

      if (not v_IsGiba or (p_Context == A_2_1_A7)) {

        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

        v_Port_us := v_Protected.Port_us;

        v_Port_ps := v_Protected.Port_ps;

      }

    }

    if (p_Context == A_2_1_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */

      v_RouteBodyList := f_RouteSet_RouteBodyList_RX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      v_From := f_IMS_RequestInDialog_FromHeaderRX();

      v_To := f_IMS_RequestInDialog_ToHeaderRX();

      v_CSeqValue := f_IMS_RequestInDialog_CSeqValueRX();     /* @sic R5s140235, R5s140385 change 3 MCC160 implementation sic@ */

      v_SessionId := f_IMS_PTC_ImsInfo_DialogGetSessionId();

    }

    else {                                       /* normal case: no re-invite */

      // route body acc. to conditions in A.2.1

      v_Pcscf := ?;                                                                      /* @sic R5s140934 change 6: P-CSCF address is checked later on sic@ */

      if (match(p_Context, (A_2_1_A6, A_2_1_A7))) {

        v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort(v_Pcscf, v_Port_ps));

      } else {

        v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(v_Pcscf, v_Port_ps));  /* @sic R5s140873: v_Pcscf;  R5s140934 change 6 sic@ */

      }
      if (not match(p_Context, (A_2_1_A6, A_2_1_A7))) {  // => no emergency call

        v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Scscf));

      }

      if (p_A8) {

        v_MessageHeader_Invite.geolocation := cr_GeolocationAny;

        v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

        v_MediaType := "multipart/mixed";

      }

      v_CallId := cr_CallId(?);

      v_From := cr_From(v_DisplayName, v_SipUrlFrom);

      v_To := cr_ToDef(p_ExpectedCalleeUri);

      v_SessionId := f_IMS_SessionId_RX_IfSupported(p_A9);

    }

    v_MessageHeader_Invite.via             := f_IMS_InviteRequest_ViaRX();                            /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address

                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */

    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := v_From;                                                 /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */

    v_MessageHeader_Invite.toField         := v_To;                                                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */

    v_MessageHeader_Invite.callId          := v_CallId;                                               /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */

    v_MessageHeader_Invite.sessionId       := v_SessionId;                                            /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Invite.cSeq            := cr_CseqDef(v_CSeqValue, "INVITE");                      /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */

    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);

    v_MessageHeader_Invite.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us));           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check of feature-param for A3, A6, A10, A12 */

    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);

    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;

    // additional headers for specific conditions:

    if (v_IsGiba or (p_Context == A_2_1_A6)) {

      v_MessageHeader_Invite.securityVerify  := omit;

      if (p_Context == A_2_1_A6) {

        v_MessageHeader_Invite.require       := omit;

        v_MessageHeader_Invite.proxyRequire  := omit;

      }

    } else {   // -> A1 or A7

      v_MessageHeader_Invite.securityVerify  := f_IMS_PTC_BuildSecurityVerifyHeader();

      if (p_Context != A_2_1_A5) {  /* @sic R5s140017 change 9.3 sic@ */

        v_MessageHeader_Invite.require       := cr_RequireSecAgree;

        v_MessageHeader_Invite.proxyRequire  := cr_ProxyRequireDef;

      }

    }

    if (not v_IsGiba) {

      v_MessageHeader_Invite.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString());

    }

    if (p_Context == A_2_1_A4) {  /* @sic R5s140017 change 9.3 sic@ */

      v_AcceptBodyList := {cr_AcceptBody("application/sdp"), cr_AcceptBody("application/3gpp-ims+xml")};

    }

    if (pc_IMS_CS_PS_SRVCCMidCall) { // @sic R5-145746 sic@

      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.mid-call+xml"); // add another element in the array

    }

    if (pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@

      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.state-and-event-info+xml"); // add another element in the array

      v_MessageHeader_Invite.recvInfo        := cr_RecvInfo ({cr_InfoPackage_Type("+g.3gpp.state-and-event")});  // @sic R5-145746 sic@

    }

    if (lengthof(v_AcceptBodyList) > 0) {

      v_MessageHeader_Invite.accept := cr_Accept(superset(all from v_AcceptBodyList));

    }

    if (p_A8) {

      v_MessageHeader_Invite.geolocation     := cr_GeolocationAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks @sic R5-133151 sic@

    }

    if (p_Context == A_2_1_A7) {

      v_MessageHeader_Invite.pPreferredID := cr_PPreferredID(?);    // Checked in f_IMS_A_2_1_Invite_CommonChecks

    }

    if (p_A3 and (p_Context == A_2_1_A4)) {                          /* @sic R5s140263 change 5, modified by R5-142949 (-> A4 only) sic@ */

      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_FeatureParamValue_3gpp_icsi_ref);   /* @sic R5s140565 change 2 sic@ */

    }

    if ((p_A3 and (p_Context == A_2_1_A4)) or p_A11) {               /* @sic R5-142949 sic@ */

      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)

    }

    return v_MessageHeader_Invite;

  }     
