
	Reason for change:
	1. In the function f_IMS_OtherResponse_200_MessageHeaderTX() a new parameter p_EmergencyType has been added. If this is set to ASSERTED_EMERGENCY but the associated parameter p_EmergencyTelUri is not present then the function generates a fatal error. Otherwise the p-assertedId header is added to the message in accordance with condition A6 of 34.229-1 A.3.1. But, condition A6 states “Response sent by SS for INVITE for emergency call”. It follows that the value ASSERTED_EMERGENCY should not be used when generating 200 OK response for other requests. In the context of test case 19.3.1, this value is passed in when generating 200 OK responses for PRACK and UPDATE requests, causing a fatal error.
2. Currently in the test case 19.3.1, the expected content-type header in the UPDATE requests is always set to “multipart/mixed”. This should only be the case when geolocation functionality is supported by the UE.

	
	

	Summary of change:
	1. In the function f_IMS_MOCallSetup_AnnexC21C25_Steps5_8(), pass the value NO_EMERGENCY when generating 200 OK responses to PRACK and UPDATE requests.
2. Remove the setting of expected context type for 19.3.1. Note : a prose CR on 34.229-1 associated with this change will be raised at RAN5#69.

Change 1 
	Function name
	f_IMS_MOCallSetup_AnnexC21C25_Steps5_8()

	Reason for change
	1. In the function f_IMS_OtherResponse_200_MessageHeaderTX() a new parameter p_EmergencyType has been added. If this is set to ASSERTED_EMERGENCY but the associated parameter p_EmergencyTelUri is not present then the function generates a fatal error. Otherwise the p-assertedId header is added to the message in accordance with condition A6 of 34.229-1 A.3.1. But, condition A6 states “Response sent by SS for INVITE for emergency call”. It follows that the value ASSERTED_EMERGENCY should not be used when generating 200 OK response for other requests. In the context of test case 19.3.1, this value is passed in when generating 200 OK responses for PRACK and UPDATE requests, causing a fatal error.

2. Currently in the test case 19.3.1, the expected content-type header in the UPDATE requests is always set to “multipart/mixed”. This should only be the case when geolocation functionality is supported by the UE.

	Summary of change
	1. In the function f_IMS_MOCallSetup_AnnexC21C25_Steps5_8(), pass the value NO_EMERGENCY when generating 200 OK responses to PRACK and UPDATE requests.

2. Remove the setting of expected context type for 19.3.1. Note : a prose CR associated with this change will be raised at RAN5#69.

	TTCN module
	IMS_Procedures_CallControl.ttcn

	MCC160 Comment
	


Before change
	function f_IMS_MOCallSetup_AnnexC21C25_Steps5_8(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                                  template (value) MessageHeader p_MessageHeader_Prev183Response,

                                                  boolean p_GeolocationInfoAvailable := false,

                                                  template SDP_Message p_SDP_Message_Step5or7 := omit,

                                                  template (omit) SDP_Message p_PrevSDP_Message := omit,

                                                  IMS_Emergency_Type p_EmergencyType := NO_EMERGENCY,

                                                  boolean p_IsCAT := false,

                                                  boolean p_IsC44 := false) runs on IMS_PTC return SDP_Message

  { /* p_SDP_Message_Step5or7 .. used in case of video (C.25) */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5w140112: p_InviteRequest changed to p_InviteRequestWithSdp */

    /* @sic R5s140686, R5-144755: parameter p_PrevSDP_Message sic@ */

    /* @sic R5-151953 new parameter: p_IsCAT sic@ */

    /* @sic R5-151968 new parameter: p_IsC44 sic@ */

    /* @sic R5s150692 change 16: parameter p_EmergencyType sic@ */

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_MessageHeader_Prev183Response);    /* @sic R5s130333 change 8.2 - acc. to A.2.4, A.2.5 same as in previous response sic@ */

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var SDP_Message v_SDP_Message := p_InviteRequestWithSdp.SdpOffer;  /* @sic R5w140112 sic@ */

    var template (present) SDP_Message v_SDP_Message_Step5or7;

    var template (present) MessageBody v_MessageBody;

    var boolean v_MediaConfigured := false;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (present) MessageHeader v_MessageHeader_Prack;

    var template (present) MessageHeader v_MessageHeader_Update;

    var template (omit) MessageBody v_MessageBodyTx := omit;

    var template (omit) ContentType v_ContentType := omit;

    var PRACK_Request v_PrackRequest;

    var UPDATE_Request v_UpdateRequest;

    var SDP_Message v_SDP_LatestOffer;

    if (isvalue(p_PrevSDP_Message)) {   /* e.g. in case of C.27 the INVITE's SDP message is not the latest SDP sent by the UE */

      v_SDP_Message := valueof(p_PrevSDP_Message);

    }

    if (not ispresent(p_SDP_Message_Step5or7)) {

      v_SDP_Message_Step5or7 := f_IMS_BuildSDP_AnnexC21_Step5or7(v_SDP_Message, p_IsCAT, p_IsC44);  /* => use speech per default */

    } else {

      v_SDP_Message_Step5or7 := p_SDP_Message_Step5or7;         /* In case of video the SDP message needs to be handed over

                                                                   NOTE: compiler warning cannot be avoided here */

    }

    // Step 5. Receive PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, p_MessageHeader_Prev183Response);

    IMS_Server.receive(car_IMS_Prack_Request(cr_PRACK_Request(v_ContactUrl, v_MessageHeader_Prack, cr_MessageBody_SDP ifpresent))) -> value v_IMS_DATA_REQ;

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    // fl_IMS_Params_CheckSigComp(v_PrackRequest.msgHeader.route.routeBody[0].rrParam);     // acc. to 34.229-1 cl. 13.2.4 step 4 (PRACK) "comp=sigcomp" is optional for Via and Route

    // Step 6. Send 200 OK

    if (ispresent(v_PrackRequest.messageBody)) {

      v_SDP_LatestOffer := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_PrackRequest.messageBody, v_SDP_Message_Step5or7);  // @sic R5w140112 sic@

      // check session version increased @sic R5-150702: done in f_IMS_SIP_DecodeMatchAndCheckSDP sic@ */

      v_ContentType := cs_ContentTypeSDP;

      v_MessageBodyTx := cs_MessageBody_SDP(f_IMS_BuildSDP_AnnexC21C25_Step6or8(v_SDP_LatestOffer));    /* @sic R5w140112 sic@ */

      v_MediaConfigured := true;

    }

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200,

                                                                   f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader, v_ContentType, -, -, tsc_OptionTagList_precondition, p_EmergencyType),  /* @sic R5s150237 change 2 sic@ @sic R5s150692 change 16 sic@ */

                                                                   v_MessageBodyTx)));

    // Step 7 Optional. Receive Update

    // Note: UE will send Update if Media is not yet configured

    if (not v_MediaConfigured) {

      v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderRX(v_InviteRequest, p_EmergencyType);  /* @sic R5s150692 change 16 sic@ */

      v_MessageBody := f_SIP_MessageBodySDP_RX(p_GeolocationInfoAvailable);

      if (p_GeolocationInfoAvailable) {

        v_MessageHeader_Update.geolocation := cr_GeolocationAny;

        v_MessageHeader_Update.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

        v_MessageHeader_Update.contentType := cr_ContentType("multipart/mixed");

      }

      if (testcasename() == "TC_19_3_1") {   // @sic !!!! R5-153760: DIRTY HACK - PROSE CR REQUIRED to correct Content-Type !!!! sic@

        v_MessageHeader_Update.contentType := cr_ContentType("multipart/mixed");

      }
      IMS_Server.receive(car_IMS_Update_Request(cr_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, v_MessageBody))) -> value v_IMS_DATA_REQ;

      v_UpdateRequest := v_IMS_DATA_REQ.Request.Update;

      v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

      v_SDP_LatestOffer := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_UpdateRequest.messageBody, v_SDP_Message_Step5or7);  /* @sic R5w140112 sic@ */

      //fl_IMS_Params_CheckSigComp(v_UpdateRequest.msgHeader.route.routeBody[0].rrParam);     // acc. to 34.229-1 cl. 13.2.4 step 6 (UPDATE) "comp=sigcomp" is optional for Via and Route

      // check session version increased @sic R5-150702: done in f_IMS_SIP_DecodeMatchAndCheckSDP sic@ */

      v_ContentType := cs_ContentTypeSDP;

      v_MessageBodyTx := cs_MessageBody_SDP(f_IMS_BuildSDP_AnnexC21C25_Step6or8(v_SDP_LatestOffer));    /*  @sic R5w140112 sic@ */

      if (testcasename() == "TC_19_3_1") {   // @sic !!!! R5-153760: DIRTY HACK - PROSE CR REQUIRED to remove checking of Route_header from UPDATE !!!! sic@ */

        f_IMS_MessageHeader_CheckContactAddr(v_UpdateRequest.msgHeader);  // @sic R5-153760: check route header sic@

      }

      // Step 8 Optional. Send 200 OK

      IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200,

                                                                     f_IMS_OtherResponse_200_MessageHeaderTX(v_UpdateRequest.msgHeader, v_ContentType, -, -, tsc_OptionTagList_precondition, p_EmergencyType),  /* @sic R5s150237 change 2 sic@ @sic R5s150692 change 16 sic@ */

                                                                     v_MessageBodyTx)));

    }

    return v_SDP_LatestOffer;

  }


After change

	function f_IMS_MOCallSetup_AnnexC21C25_Steps5_8(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                                  template (value) MessageHeader p_MessageHeader_Prev183Response,

                                                  boolean p_GeolocationInfoAvailable := false,

                                                  template SDP_Message p_SDP_Message_Step5or7 := omit,

                                                  template (omit) SDP_Message p_PrevSDP_Message := omit,

                                                  IMS_Emergency_Type p_EmergencyType := NO_EMERGENCY,

                                                  boolean p_IsCAT := false,

                                                  boolean p_IsC44 := false) runs on IMS_PTC return SDP_Message

  { /* p_SDP_Message_Step5or7 .. used in case of video (C.25) */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5w140112: p_InviteRequest changed to p_InviteRequestWithSdp */

    /* @sic R5s140686, R5-144755: parameter p_PrevSDP_Message sic@ */

    /* @sic R5-151953 new parameter: p_IsCAT sic@ */

    /* @sic R5-151968 new parameter: p_IsC44 sic@ */

    /* @sic R5s150692 change 16: parameter p_EmergencyType sic@ */

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_MessageHeader_Prev183Response);    /* @sic R5s130333 change 8.2 - acc. to A.2.4, A.2.5 same as in previous response sic@ */

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var SDP_Message v_SDP_Message := p_InviteRequestWithSdp.SdpOffer;  /* @sic R5w140112 sic@ */

    var template (present) SDP_Message v_SDP_Message_Step5or7;

    var template (present) MessageBody v_MessageBody;

    var boolean v_MediaConfigured := false;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (present) MessageHeader v_MessageHeader_Prack;

    var template (present) MessageHeader v_MessageHeader_Update;

    var template (omit) MessageBody v_MessageBodyTx := omit;

    var template (omit) ContentType v_ContentType := omit;

    var PRACK_Request v_PrackRequest;

    var UPDATE_Request v_UpdateRequest;

    var SDP_Message v_SDP_LatestOffer;

    if (isvalue(p_PrevSDP_Message)) {   /* e.g. in case of C.27 the INVITE's SDP message is not the latest SDP sent by the UE */

      v_SDP_Message := valueof(p_PrevSDP_Message);

    }

    if (not ispresent(p_SDP_Message_Step5or7)) {

      v_SDP_Message_Step5or7 := f_IMS_BuildSDP_AnnexC21_Step5or7(v_SDP_Message, p_IsCAT, p_IsC44);  /* => use speech per default */

    } else {

      v_SDP_Message_Step5or7 := p_SDP_Message_Step5or7;         /* In case of video the SDP message needs to be handed over

                                                                   NOTE: compiler warning cannot be avoided here */

    }

    // Step 5. Receive PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, p_MessageHeader_Prev183Response);

    IMS_Server.receive(car_IMS_Prack_Request(cr_PRACK_Request(v_ContactUrl, v_MessageHeader_Prack, cr_MessageBody_SDP ifpresent))) -> value v_IMS_DATA_REQ;

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    // fl_IMS_Params_CheckSigComp(v_PrackRequest.msgHeader.route.routeBody[0].rrParam);     // acc. to 34.229-1 cl. 13.2.4 step 4 (PRACK) "comp=sigcomp" is optional for Via and Route

    // Step 6. Send 200 OK

    if (ispresent(v_PrackRequest.messageBody)) {

      v_SDP_LatestOffer := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_PrackRequest.messageBody, v_SDP_Message_Step5or7);  // @sic R5w140112 sic@

      // check session version increased @sic R5-150702: done in f_IMS_SIP_DecodeMatchAndCheckSDP sic@ */

      v_ContentType := cs_ContentTypeSDP;

      v_MessageBodyTx := cs_MessageBody_SDP(f_IMS_BuildSDP_AnnexC21C25_Step6or8(v_SDP_LatestOffer));    /* @sic R5w140112 sic@ */

      v_MediaConfigured := true;

    }

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200,

                                                                   f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader, v_ContentType, -, -, tsc_OptionTagList_precondition, NO_EMERGENCY), /* @sic R5s150237 change 2 sic@ @sic R5s150692 change 16 sic@ */

                                                                   v_MessageBodyTx)));

    // Step 7 Optional. Receive Update

    // Note: UE will send Update if Media is not yet configured

    if (not v_MediaConfigured) {

      v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderRX(v_InviteRequest, p_EmergencyType);  /* @sic R5s150692 change 16 sic@ */

      v_MessageBody := f_SIP_MessageBodySDP_RX(p_GeolocationInfoAvailable);

      if (p_GeolocationInfoAvailable) {

        v_MessageHeader_Update.geolocation := cr_GeolocationAny;

        v_MessageHeader_Update.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

        v_MessageHeader_Update.contentType := cr_ContentType("multipart/mixed");

      }

      /* REMOVED if (testcasename() == "TC_19_3_1") {   // @sic !!!! R5-153760: DIRTY HACK - PROSE CR REQUIRED to correct Content-Type !!!! sic@

         v_MessageHeader_Update.contentType := cr_ContentType("multipart/mixed");

      }*/
      IMS_Server.receive(car_IMS_Update_Request(cr_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, v_MessageBody))) -> value v_IMS_DATA_REQ;

      v_UpdateRequest := v_IMS_DATA_REQ.Request.Update;

      v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

      v_SDP_LatestOffer := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_UpdateRequest.messageBody, v_SDP_Message_Step5or7);  /* @sic R5w140112 sic@ */

      //fl_IMS_Params_CheckSigComp(v_UpdateRequest.msgHeader.route.routeBody[0].rrParam);     // acc. to 34.229-1 cl. 13.2.4 step 6 (UPDATE) "comp=sigcomp" is optional for Via and Route

      // check session version increased @sic R5-150702: done in f_IMS_SIP_DecodeMatchAndCheckSDP sic@ */

      v_ContentType := cs_ContentTypeSDP;

      v_MessageBodyTx := cs_MessageBody_SDP(f_IMS_BuildSDP_AnnexC21C25_Step6or8(v_SDP_LatestOffer));    /*  @sic R5w140112 sic@ */

      if (testcasename() == "TC_19_3_1") {   // @sic !!!! R5-153760: DIRTY HACK - PROSE CR REQUIRED to remove checking of Route_header from UPDATE !!!! sic@ */

        f_IMS_MessageHeader_CheckContactAddr(v_UpdateRequest.msgHeader);  // @sic R5-153760: check route header sic@

      }

      // Step 8 Optional. Send 200 OK

      IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200,

                                                                     f_IMS_OtherResponse_200_MessageHeaderTX(v_UpdateRequest.msgHeader, v_ContentType, -, -, tsc_OptionTagList_precondition, NO_EMERGENCY),  /* @sic R5s150237 change 2 sic@ @sic R5s150692 change 16 sic@ */

                                                                     v_MessageBodyTx)));

    }

    return v_SDP_LatestOffer;

  }


